contract BCC is Context, IERC20, Ownable {
    using SafeMath for uint256;
    using Address for address;

    IUniswapV2Router02 public uniswapV2Router;

    mapping(address => uint256) private _rOwned;
    mapping(address => uint256) private _tOwned;
    mapping(address => mapping(address => uint256)) private _allowances;
    mapping(address => bool) private _isExcludedFromFee;
    mapping(address => bool) private _isExcluded;
    mapping (address => bool) private _isBlackhat;

    string private _name = "BlueChip Capital Token";
    string private _symbol = "BCC";
    uint8 private _decimals = 9;

    uint256 private constant MAX = ~uint256(0);
    uint256 private _tTotal = 100000000 * 10**9; //100M 
    uint256 private _rTotal = (MAX - (MAX % _tTotal));
    uint256 private _tFeeTotal;
    uint256 private _tBurnTotal;

    // Transfer Fees
    uint256 public _liquidityFee = 3;
    uint256 public _taxFee = 2;
    uint256 public _burnFee = 1;

    uint256 private _previousTaxFee = _taxFee;
    uint256 private _previousBurnFee = _burnFee;
    uint256 private _previousLiquidityFee = _liquidityFee;
    uint256 private _sellMultiplier = 3;
    uint256 private _maxTxAmount = 500000 * 10**9; //500K
    uint256 private numTokensSellToAddToLiquidity = 100000 * 10**9; //100K
    uint256 private _liqAddBlock;
    uint256 private snipeBlockAmt;
    uint256 public blackhatsCaught;
    
    address public uniswapV2Pair;
    address payable public _marketingWalletAddress = 0xaCC32d650a908e98E5a02fc0a42A18B2860c5882;
    address payable public _treasuryWalletAddress = 0x6311356734Bca619F5627BcC3238a5134A65d5f3;
    address public sweepablePair;

    bool inSwapAndLiquify;
    bool public swapAndLiquifyEnabled = true;
    bool public _hasLiqBeenAdded = false;
    
    event BlackhatCaught(address blackhatAddress);
    event TakeFeesUpdated(bool takeFee);
    event MinTokensBeforeSwapUpdated(uint256 minTokensBeforeSwap);
    event SwapAndLiquifyEnabledUpdated(bool enabled);
    event SwapAndLiquify(uint256 tokensSwapped, uint256 ethReceived);
    event SetNumTokensSellToAddToLiquidity(uint256 numTokensSellToAddToLiquidity);
    event SetMaxTxAmount(uint256 maxTxAmount);
    event SetBurnFee(uint256 _previousBurnFee, uint256 newBurnFee);
    event SetTaxFee(uint256 _previousTaxFee, uint256 newTaxFee);
    event SetLiquidityFee(uint256 _previousLiquidityFee, uint256 newLiquidityFee);
    event SetMarketingWallet(address newMarketingWallet);
    event SetTreasuryWallet(address newTreasuryWallet);
    event SetTreasuryMarketingDistribution(uint256 TreasuryDistribution, uint256 MarketingDistribution);
    event ReceiveFallback(address,uint256);

    modifier lockTheSwap {
        inSwapAndLiquify = true;
        _;
        inSwapAndLiquify = false;
    }

    modifier onlyPair() {
        require(
            msg.sender == uniswapV2Pair,
            "Empire::onlyPair: Insufficient Privileges"
        );
        _;
    }

    address[] private _excluded;

    constructor(uint256 _snipeBlockAmt) public {
        
        _rOwned[_msgSender()] = _rTotal;

        //exclude owner and this contract from fee
        _isExcludedFromFee[owner()] = true;
        _isExcludedFromFee[address(this)] = true;

        IUniswapV2Router02 _uniswapRouter = IUniswapV2Router02(0x7a250d5630B4cF539739dF2C5dAcb4c659F2488D);
        uniswapV2Pair = IUniswapV2Factory(_uniswapRouter.factory())
        .createPair(address(this), _uniswapRouter.WETH());

        uniswapV2Router = _uniswapRouter;
        snipeBlockAmt = _snipeBlockAmt;

        _isExcludedFromFee[owner()] = true;
        _isExcludedFromFee[address(this)] = true;  
        _isExcludedFromFee[_marketingWalletAddress] = true;  
        _isExcludedFromFee[_treasuryWalletAddress] = true;  

        _isExcluded[uniswapV2Pair] = true;
        _excluded.push(uniswapV2Pair);

        emit Transfer(address(0), _msgSender(), _tTotal);
    }

    receive() external payable {
        emit ReceiveFallback(msg.sender, msg.value);
    }

    function name() public view returns (string memory) {
        return _name;
    }

    function symbol() public view returns (string memory) {
        return _symbol;
    }

    function decimals() public view returns (uint8) {
        return _decimals;
    }

    function totalSupply() public view override returns (uint256) {
        return _tTotal;
    }

    function balanceOf(address account) public view override returns (uint256) {
        if (_isExcluded[account]) return _tOwned[account];
        return tokenFromReflection(_rOwned[account]);
    }

    function transfer(address recipient, uint256 amount) public override returns (bool) {
        _transfer(_msgSender(), recipient, amount);
        return true;
    }

    function allowance(address owner, address spender) public view override returns (uint256)
    {
        return _allowances[owner][spender];
    }

    function approve(address spender, uint256 amount) public override returns (bool)
    {
        _approve(_msgSender(), spender, amount);
        return true;
    }

    function transferFrom(address sender, address recipient, uint256 amount) public override returns (bool) {
        _transfer(sender, recipient, amount);
        _approve(
            sender,
            _msgSender(),
            _allowances[sender][_msgSender()].sub(
                amount,
                "ERC20: transfer amount exceeds allowance"
            )
        );
        return true;
    }

    function increaseAllowance(address spender, uint256 addedValue) public virtual returns (bool) {
        _approve(
            _msgSender(),
            spender,
            _allowances[_msgSender()][spender].add(addedValue)
        );
        return true;
    }

    function decreaseAllowance(address spender, uint256 subtractedValue) public virtual returns (bool) {
        _approve(
            _msgSender(),
            spender,
            _allowances[_msgSender()][spender].sub(
                subtractedValue,
                "ERC20: decreased allowance below zero"
            )
        );
        return true;
    }

    function isExcludedFromReward(address account) public view returns (bool) {
        return _isExcluded[account];
    }

    function totalFees() public view returns (uint256) {
        return _tFeeTotal;
    }

    function totalBurn() public view returns (uint256) {
        return _tBurnTotal;
    }

    function getCirculatingSupply() public view returns (uint256) {
        uint256 burnAddressTokens = balanceOf(0x000000000000000000000000000000000000dEaD).add(balanceOf(0x0000000000000000000000000000000000000000));
        uint256 removedAddressTokens;

        for (uint256 i = 0; i < removedFromCirculation.length; i++) {
            removedAddressTokens += balanceOf(removedFromCirculation[i]);
        }

        return totalSupply().sub(burnAddressTokens).sub(removedAddressTokens);
    }

    function reflectionFromToken(uint256 tAmount, bool deductTransferFee) public view returns (uint256) {
        require(tAmount <= _tTotal, "Amount must be less than supply");
        if (!deductTransferFee) {
            (uint256 rAmount, , , , , , ) = _getValues(tAmount);
            return rAmount;
        } else {
            (, uint256 rTransferAmount, , , , , ) = _getValues(tAmount);
            return rTransferAmount;
        }
    }

    function tokenFromReflection(uint256 rAmount) public view returns (uint256) {
        require(
            rAmount <= _rTotal,
            "Amount must be less than total reflections"
        );
        uint256 currentRate = _getRate();
        return rAmount.div(currentRate);
    }

    function _transferBothExcluded(address sender, address recipient, uint256 tAmount) private {
        uint256 currentRate = _getRate();
        (
            uint256 rAmount,
            uint256 rTransferAmount,
            uint256 rFee,
            uint256 tTransferAmount,
            uint256 tFee,
            uint256 tLiquidity,
            uint256 tBurn
        ) = _getValues(tAmount);
        uint256 rBurn = tBurn.mul(currentRate);
        _tOwned[sender] = _tOwned[sender].sub(tAmount);
        _rOwned[sender] = _rOwned[sender].sub(rAmount);
        _tOwned[recipient] = _tOwned[recipient].add(tTransferAmount);
        _rOwned[recipient] = _rOwned[recipient].add(rTransferAmount);
        _takeLiquidity(tLiquidity);
        _reflectFee(rFee, rBurn, tFee, tBurn);
        emit Transfer(sender, recipient, tTransferAmount);
    }

    function _reflectFee(uint256 rFee, uint256 rBurn, uint256 tFee, uint256 tBurn) private {
        _rTotal = _rTotal.sub(rFee).sub(rBurn);
        _tBurnTotal = _tBurnTotal.add(tBurn);
        _tFeeTotal = _tFeeTotal.add(tFee);
        _tTotal = _tTotal.sub(tBurn);
    }

    function _getValues(uint256 tAmount) private view returns (uint256, uint256, uint256, uint256, uint256, uint256, uint256) {
        (
            uint256 tTransferAmount,
            uint256 tFee,
            uint256 tLiquidity,
            uint256 tBurn
        ) = _getTValues(tAmount);
        (uint256 rAmount, uint256 rTransferAmount, uint256 rFee) =
            _getRValues(tAmount, tFee, tLiquidity, tBurn, _getRate());
        return (
            rAmount,
            rTransferAmount,
            rFee,
            tTransferAmount,
            tFee,
            tLiquidity,
            tBurn
        );
    }

    function _getTValues(uint256 tAmount) private view returns (uint256, uint256, uint256, uint256) {
        uint256 tFee = calculateTaxFee(tAmount);
        uint256 tLiquidity = calculateLiquidityFee(tAmount);
        uint256 tBurn = calculateBurnFee(tAmount);
        uint256 tTransferAmount = tAmount.sub(tFee).sub(tLiquidity).sub(tBurn);
        require(tTransferAmount > tFee.add(tLiquidity).add(tBurn), 'Amount is less than fees');

        return (tTransferAmount, tFee, tLiquidity, tBurn);
    }

    function _getRValues(uint256 tAmount, uint256 tFee, uint256 tLiquidity, uint256 tBurn, uint256 currentRate) private pure returns (uint256, uint256, uint256) {
        uint256 rAmount = tAmount.mul(currentRate);
        uint256 rFee = tFee.mul(currentRate);
        uint256 rLiquidity = tLiquidity.mul(currentRate);
        uint256 rBurn = tBurn.mul(currentRate);
        uint256 rTransferAmount = rAmount.sub(rFee).sub(rLiquidity).sub(rBurn);
        return (rAmount, rTransferAmount, rFee);
    }

    function _getRate() private view returns (uint256) {
        (uint256 rSupply, uint256 tSupply) = _getCurrentSupply();
        return rSupply.div(tSupply);
    }

    function _getCurrentSupply() private view returns (uint256, uint256) {
        uint256 rSupply = _rTotal;
        uint256 tSupply = _tTotal;
        for (uint256 i = 0; i < _excluded.length; i++) {
            if (
                _rOwned[_excluded[i]] > rSupply ||
                _tOwned[_excluded[i]] > tSupply
            ) return (_rTotal, _tTotal);
            rSupply = rSupply.sub(_rOwned[_excluded[i]]);
            tSupply = tSupply.sub(_tOwned[_excluded[i]]);
        }
        if (rSupply < _rTotal.div(_tTotal)) return (_rTotal, _tTotal);
        return (rSupply, tSupply);
    }

    function _takeLiquidity(uint256 tLiquidity) private {
        uint256 currentRate = _getRate();
        uint256 rLiquidity = tLiquidity.mul(currentRate);
        _rOwned[address(this)] = _rOwned[address(this)].add(rLiquidity);
        if (_isExcluded[address(this)])
            _tOwned[address(this)] = _tOwned[address(this)].add(tLiquidity);
    }

    function calculateTaxFee(uint256 _amount) private view returns (uint256) {
        return _amount.mul(_taxFee).div(100);
    }

    function calculateBurnFee(uint256 _amount) private view returns (uint256) {
        return _amount.mul(_burnFee).div(100);
    }

    function calculateLiquidityFee(uint256 _amount) private view returns (uint256) {
        return _amount.mul(_liquidityFee).div(100);
    }

    function removeAllFee() private {
        if (_taxFee == 0 && _liquidityFee == 0 && _burnFee == 0) return;

        _previousTaxFee = _taxFee;
        _previousBurnFee = _burnFee;
        _previousLiquidityFee = _liquidityFee;

        _taxFee = 0;
        _burnFee = 0;
        _liquidityFee = 0;
    }

    function restoreAllFee() private {
        _taxFee = _previousTaxFee;
        _burnFee = _previousBurnFee;
        _liquidityFee = _previousLiquidityFee;
    }

    function isExcludedFromFee(address account) public view returns (bool) {
        return _isExcludedFromFee[account];
    }

    function _approve(
        address owner,
        address spender,
        uint256 amount
    ) private {
        require(owner != address(0), "ERC20: approve from the zero address");
        require(spender != address(0), "ERC20: approve to the zero address");
        _allowances[owner][spender] = amount;
        emit Approval(owner, spender, amount);
    }

    function _transfer(
        address from,
        address to,
        uint256 amount
    ) private {
        require(from != address(0), "ERC20: transfer from the zero address");
        require(to != address(0), "ERC20: transfer to the zero address");
        require(amount > 0, "Transfer amount must be greater than zero");
        require(!_isBlackhat[from], "blackhats can not transfer");
        require(!_isBlackhat[to], "blackhats can not transfer");
        require(!_isBlackhat[msg.sender], "blackhats can not transfer");

        if (!_hasLiqBeenAdded) {
            _checkLiquidityAdd(to);
        } else {
            if (_liqAddBlock > 0 
                && from == uniswapV2Pair
            ) {
                if (block.number - _liqAddBlock < snipeBlockAmt) {
                    _isBlackhat[to] = true;
                    excludeFromReward(to);

                    blackhatsCaught ++;
                    emit BlackhatCaught(to); //pow
                }
            }
        }
        
        if (from != owner() && to != owner() && !inSwapAndLiquify) { 
            require(
                amount <= _maxTxAmount,
                "Transfer amount exceeds the maxTxAmount."
            );
        }

        // is the token balance of this contract address over the min number of
        // tokens that we need to initiate a swap + liquidity lock?
        // also, don't get caught in a circular liquidity event.
        // also, don't swap & liquify if sender is uniswap pair.
        uint256 contractTokenBalance = balanceOf(address(this));

        if (contractTokenBalance >= _maxTxAmount) {
            contractTokenBalance = _maxTxAmount;
        }

        bool overMinTokenBalance =
            contractTokenBalance >= numTokensSellToAddToLiquidity;
        if (
            overMinTokenBalance &&
            !inSwapAndLiquify &&
            from != uniswapV2Pair &&
            swapAndLiquifyEnabled
        ) {
            contractTokenBalance = numTokensSellToAddToLiquidity;
            swapAndLiquify(contractTokenBalance);
        }

        //indicates if fee should be deducted from transfer
        bool takeFee = true;

        //if any account belongs to _isExcludedFromFee account then remove the fee
        if (_isExcludedFromFee[from] || _isExcludedFromFee[to]) {
            takeFee = false;
        }

        if (takeFee && to == uniswapV2Pair && !inSwapAndLiquify) {
            _liquidityFee = _liquidityFee.mul(_sellMultiplier);
        }

        //transfer amount, it will take tax, burn, liquidity fee
        _tokenTransfer(from, to, amount, takeFee);

        restoreAllFee();
    }

    function swapAndLiquify(uint256 contractTokenBalance) private lockTheSwap {

        // capture the contract's current ETH balance.
        // this is so that we can capture exactly the amount of ETH that the
        // swap creates, and not make the liquidity event include any ETH that
        // has been manually sent to the contract
        uint256 initialBalance = address(this).balance;

        // swap tokens for ETH
        swapTokensForEth(contractTokenBalance);

        // how much ETH did we just swap into?
        uint256 newBalance = address(this).balance.sub(initialBalance);

        sendETHToCapitalFund(newBalance);
        
        emit SwapAndLiquify(contractTokenBalance, newBalance);
    }

    function swapTokensForEth(uint256 tokenAmount) private {
        // generate the uniswap pair path of token -> weth
        address[] memory path = new address[](2);
        path[0] = address(this);
        path[1] = uniswapV2Router.WETH();

        _approve(address(this), address(uniswapV2Router), tokenAmount);

        // make the swap
        uniswapV2Router.swapExactTokensForETHSupportingFeeOnTransferTokens(
            tokenAmount,
            0, // accept any amount of ETH
            path,
            address(this),
            block.timestamp
        );
    }

    function sendETHToCapitalFund(uint256 newBalance) private {
        _treasuryWalletAddress.transfer(newBalance.div(2)); 
        _marketingWalletAddress.transfer(newBalance.div(2)); 
    }

    //this method is responsible for taking all fee, if takeFee is true
    function _tokenTransfer(address sender, address recipient, uint256 amount, bool takeFee) private {
        if (!takeFee) removeAllFee();

        if (_isExcluded[sender] && !_isExcluded[recipient]) {
            _transferFromExcluded(sender, recipient, amount);
        } else if (!_isExcluded[sender] && _isExcluded[recipient]) {
            _transferToExcluded(sender, recipient, amount);
        } else if (!_isExcluded[sender] && !_isExcluded[recipient]) {
            _transferStandard(sender, recipient, amount);
        } else if (_isExcluded[sender] && _isExcluded[recipient]) {
            _transferBothExcluded(sender, recipient, amount);
        } else {
            _transferStandard(sender, recipient, amount);
        }

        if (!takeFee) restoreAllFee();
    }

    function _transferStandard(address sender, address recipient, uint256 tAmount) private {
        uint256 currentRate = _getRate();
        (
            uint256 rAmount,
            uint256 rTransferAmount,
            uint256 rFee,
            uint256 tTransferAmount,
            uint256 tFee,
            uint256 tLiquidity,
            uint256 tBurn
        ) = _getValues(tAmount);
        uint256 rBurn = tBurn.mul(currentRate);
        _rOwned[sender] = _rOwned[sender].sub(rAmount);
        _rOwned[recipient] = _rOwned[recipient].add(rTransferAmount);
        _takeLiquidity(tLiquidity);
        _reflectFee(rFee, rBurn, tFee, tBurn);
        emit Transfer(sender, recipient, tTransferAmount);
    }

    function _transferToExcluded(address sender, address recipient, uint256 tAmount) private {
        uint256 currentRate = _getRate();
        (
            uint256 rAmount,
            uint256 rTransferAmount,
            uint256 rFee,
            uint256 tTransferAmount,
            uint256 tFee,
            uint256 tLiquidity,
            uint256 tBurn
        ) = _getValues(tAmount);
        uint256 rBurn = tBurn.mul(currentRate);
        _rOwned[sender] = _rOwned[sender].sub(rAmount);
        _tOwned[recipient] = _tOwned[recipient].add(tTransferAmount);
        _rOwned[recipient] = _rOwned[recipient].add(rTransferAmount);
        _takeLiquidity(tLiquidity);
        _reflectFee(rFee, rBurn, tFee, tBurn);
        emit Transfer(sender, recipient, tTransferAmount);
    }

    function _transferFromExcluded(address sender, address recipient, uint256 tAmount) private {
        uint256 currentRate = _getRate();
        (
            uint256 rAmount,
            uint256 rTransferAmount,
            uint256 rFee,
            uint256 tTransferAmount,
            uint256 tFee,
            uint256 tLiquidity,
            uint256 tBurn
        ) = _getValues(tAmount);
        uint256 rBurn = tBurn.mul(currentRate);
        _tOwned[sender] = _tOwned[sender].sub(tAmount);
        _rOwned[sender] = _rOwned[sender].sub(rAmount);
        _rOwned[recipient] = _rOwned[recipient].add(rTransferAmount);
        _takeLiquidity(tLiquidity);
        _reflectFee(rFee, rBurn, tFee, tBurn);
        emit Transfer(sender, recipient, tTransferAmount);
    }

    // blackhat pow
    function _checkLiquidityAdd(address to) private {
        require(!_hasLiqBeenAdded, "Liquidity already added and marked.");
        if (to == uniswapV2Pair) {
            _hasLiqBeenAdded = true;
            _liqAddBlock = block.number;
        }
    }

    function isBlackhatCheck(address account) public view returns (bool) {
        return _isBlackhat[account];
    }
    
    function addBlackhat(address blackhatAddress) external onlyOwner() {
        require(blackhatAddress != uniswapV2Pair, "ERC20: Can not add uniswapV2Pair to blackhat list");
        require(blackhatAddress != address(uniswapV2Router), "ERC20: Can not add uniswapV2Router to blackhat list");

        _isBlackhat[blackhatAddress] = true;
    }
    
    function removeBlackhat(address blackhatAddress) external onlyOwner() {
        require(_isBlackhat[blackhatAddress], "ERC20: Is not blackhat");
        _isBlackhat[blackhatAddress] = false;
    }
    
    // owner only
    function setNumTokensSellToAddToLiquidity(uint256 _numTokensSellToAddToLiquidity) external onlyOwner() {
        numTokensSellToAddToLiquidity = _numTokensSellToAddToLiquidity;

        emit SetNumTokensSellToAddToLiquidity(numTokensSellToAddToLiquidity);
    }

    function setMaxTxAmount(uint256 maxTxAmount) external onlyOwner() {
        _maxTxAmount = maxTxAmount;

        emit SetMaxTxAmount(maxTxAmount);
    }

    function setSwapAndLiquifyEnabled(bool _enabled) external onlyOwner() {
        swapAndLiquifyEnabled = _enabled;
        emit SwapAndLiquifyEnabledUpdated(_enabled);
    }
        
    function setFees(uint256 burnFee, uint256 taxFee, uint256 liquidityFee, uint256 sellMultiplier) external onlyOwner() {
        require(burnFee >= 0 && burnFee <= 15 && taxFee >= 0 && taxFee <= 15 && liquidityFee >= 0 && liquidityFee <= 15, "Must be a number between 0 and 15");

        _previousBurnFee = _burnFee;
        _burnFee = burnFee;
        _previousTaxFee = _taxFee;
        _taxFee = taxFee;
        _previousLiquidityFee = _liquidityFee;
        _liquidityFee = liquidityFee;
        _sellMultiplier = sellMultiplier;
        
        emit SetBurnFee(_previousBurnFee, burnFee);
        emit SetTaxFee(_previousTaxFee, taxFee);
        emit SetLiquidityFee(_previousLiquidityFee, liquidityFee);
    }

    function excludeFromReward(address account) public onlyOwner() {
        require(!_isExcluded[account], "Account is already excluded");
        if (_rOwned[account] > 0) {
            _tOwned[account] = tokenFromReflection(_rOwned[account]);
        }
        _isExcluded[account] = true;
        _excluded.push(account);
    }

    function includeInReward(address account) external onlyOwner() {
        require(_isExcluded[account], "Account is already included");
        for (uint256 i = 0; i < _excluded.length; i++) {
            if (_excluded[i] == account) {
                _excluded[i] = _excluded[_excluded.length - 1];
                _tOwned[account] = 0;
                _isExcluded[account] = false;
                _excluded.pop();
                break;
            }
        }
    }

    function excludeFromFee(address account) public onlyOwner() {
        require(!_isExcludedFromFee[account], "Account is already excluded");
        _isExcludedFromFee[account] = true;
    }

    function includeInFee(address account) public onlyOwner() {
        require(_isExcludedFromFee[account], "Account is already included");
        _isExcludedFromFee[account] = false;
    }

    function setMarketingWallet(address payable marketingWallet) external onlyOwner() {
        _marketingWalletAddress = marketingWallet;
        emit SetMarketingWallet(marketingWallet);
    }

    function setTreasuryWallet(address payable treasuryWallet) external onlyOwner() {
        _treasuryWalletAddress = treasuryWallet;
        emit SetTreasuryWallet(treasuryWallet);
    }

    function setPair(address payable _uniswapV2Pair) external onlyOwner() {
        uniswapV2Pair = _uniswapV2Pair;
    }
    
    // Admin functions to remove tokens mistakenly sent to this address
    function transferAnyTokens(address _tokenAddr, address _to, uint256 _amount) external onlyOwner() {
        require(_tokenAddr != address(this), "Tranfer failed. Can't remove BCC");
        require(IERC20(_tokenAddr).transfer(_to, _amount), "Transfer failed");
    }

    function transferETH(address payable recipient, uint256 amount) external onlyOwner()  {
        require(address(this).balance >= amount, "Address: insufficient balance");

        // solhint-disable-next-line avoid-low-level-calls, avoid-call-value
        (bool success, ) = recipient.call{ value: amount }("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    // EmpireDEX 
    event Sweep(uint256 sweepAmount);
    event Unsweep(uint256 unsweepAmount);
    event AddressRemovedFromSupply(address removedAddress);
    event AddressAddedToSupply(address addedAddress);

    address[] public removedFromCirculation;
    mapping (address=> uint256) public removedAddressIndex;

    function createSweepablePair(IEmpireFactory _factory, bool update) external onlyOwner() {
        PairType pairType =
            address(this) < uniswapV2Router.WETH()
                ? PairType.SweepableToken1
                : PairType.SweepableToken0;
        sweepablePair = _factory.createPair(uniswapV2Router.WETH(), address(this), pairType, 0);

        _isExcluded[sweepablePair] = true;
        _excluded.push(sweepablePair);

        if(update) { 
            uniswapV2Pair = sweepablePair;
        }

    }

    // Allows for Owner to sweep any tokens above the price floor
    function sweep(uint256 amount, bytes calldata data) external onlyOwner() {
        IEmpirePair(sweepablePair).sweep(amount, data);
        emit Sweep(amount);
    }

    // Sweep function utilised by the trading pair address
    function empireSweepCall(uint256 amount, bytes calldata) external onlyPair() {
        IERC20(uniswapV2Router.WETH()).transfer(owner(), amount);
    }

    // Allows for Owner to add liquidity back into the trading pair
    function unsweep(uint256 amount) external onlyOwner() {
        IERC20(uniswapV2Router.WETH()).approve(sweepablePair, amount);
        IEmpirePair(sweepablePair).unsweep(amount);
        emit Unsweep(amount);
    }

    // Removes an address from the circulating supply
    function removeAddressFromSupply(address removedAddress) external onlyOwner() {
        uint256 id = removedFromCirculation.length;
        require(removedAddress != address(uniswapV2Router), "Cannot add router (already accounted for)");
        require(removedAddress != 0x000000000000000000000000000000000000dEaD || removedAddress != 0x0000000000000000000000000000000000000000, "Cannot add burn address (already accounted for)");
        
        removedAddressIndex[removedAddress] = id;
        removedFromCirculation.push(removedAddress);
        emit AddressRemovedFromSupply(removedAddress);
    }

    // Re-Adds an address to the circulating supply
    function addAddressToSupply(address addedAddress) external onlyOwner() {
        uint256 id = removedAddressIndex[addedAddress];
        require(addedAddress != address(uniswapV2Router), "Cannot add router (already accounted for)");
        require(addedAddress != 0x000000000000000000000000000000000000dEaD || addedAddress != 0x0000000000000000000000000000000000000000, "Cannot add burn address (already accounted for)");

        delete removedFromCirculation[id];
        emit AddressAddedToSupply(addedAddress);
    }

}
function empireSweepCall(uint256 amount, bytes calldata) external onlyPair() {
    IERC20(uniswapV2Router.WETH()).transfer(owner(), amount);
}

// Sweep function utilised by the trading pair address